#-------------------------------------------------------------------------------
# Copyright (c) 2020-2021, Arm Limited. All rights reserved.
#
# SPDX-License-Identifier: BSD-3-Clause
#
#-------------------------------------------------------------------------------

cmake_policy(SET CMP0079 NEW)

add_library(tfm_psa_rot_partition_ipc_service STATIC)

target_sources(tfm_psa_rot_partition_ipc_service
    PRIVATE
        ./tfm_ipc_service_test.c
)

# The generated sources
target_sources(tfm_psa_rot_partition_ipc_service
    PRIVATE
        $<$<BOOL:${TFM_PSA_API}>:
            ${CMAKE_BINARY_DIR}/generated/test_services/tfm_ipc_service/auto_generated/intermedia_tfm_ipc_service_test.c>
)

target_sources(tfm_partitions
    INTERFACE
        $<$<BOOL:${TFM_PSA_API}>:
            ${CMAKE_BINARY_DIR}/generated/test_services/tfm_ipc_service/auto_generated/load_info_tfm_ipc_service_test.c>
)

# Set include directory
target_include_directories(tfm_psa_rot_partition_ipc_service
    PUBLIC
        .
    PRIVATE
        ${CMAKE_BINARY_DIR}/generated/test_services/tfm_ipc_service
)

target_include_directories(tfm_partitions
    INTERFACE
        ${CMAKE_BINARY_DIR}/generated/test_services/tfm_ipc_service
)

target_link_libraries(tfm_psa_rot_partition_ipc_service
    PRIVATE
        tfm_secure_api
        platform_s
        tfm_test_framework_s
        tfm_arch
)

target_link_libraries(tfm_spm
    PRIVATE
        tfm_psa_rot_partition_ipc_service
)

############################ Partition Defs ####################################

target_link_libraries(tfm_partitions
    INTERFACE
        tfm_psa_rot_partition_ipc_service
)

target_compile_definitions(tfm_partition_defs
    INTERFACE
        TFM_PARTITION_TEST_CORE_IPC
        $<$<VERSION_GREATER:${TFM_ISOLATION_LEVEL},1>:TFM_IPC_ISOLATION_2_TEST_READ_ONLY_MEM>
)
